

The MANOVA for Dependent Groups -- Analysis of 2-Within-Group Data with Two or More Quantitative DVs

Application: To compare means of two or more quantitative variables obtained from 2 dependent groups.

Research Hypothesis: The researcher hypothesized that there would be more errors and performance would be slower during constant noise than during intermittent noise.

H0: for this analysis: Performance during intermittent noise will yield the same mean error rates and mean task speed as during constant noise.

SPSS Code:

```
data list free / intererr consterr interspd constspd.

variable labels      intererr 'errors during intermittent noise condition'
                    / consterr 'errors during constant noise condition'
                    / interspd 'speed during intermittent noise condition'
                    / constspd 'speed during constant noise condition'.

begin data.
19 24 110 114
26 31 120 112
18 27 130 132
17 29 110 103
20 33 98 86
20 25 119 125
end data.

manova intererr consterr interspd constspd
      / wsfactors noise(2)
      / measures error speed
      / print cellinfo(means) signif(multiv).
```

There are four variables for each participant -- each of two DVs measured during the completion of each of two IV conditions.

The order in which the DVs are listed is important! They must be grouped with all of one DV first, then all of the second DV, etc. The order of reference to IV conditions must be the same in each DV group. In this example there are two groups of DV (first both "error" DVs and then both "speed" DVs), and within each group the DV from the "intermittent" IV condition is given first, followed by the DV from the "constant" IV condition.

The "wsfactors" subcommand is used to provide a name for the IV (and gives the number of IV conditions in parentheses). SPSS naming conventions must be followed.

The "measures" subcommand is used to provide names for the DVs (follow SPSS naming conventions). The first DV named must correspond to the first group of DVs, the second DV name must correspond to the second group of DVs, etc.

The "statistics" subcommand requests univariate statistics and the multivariate test of mean differences between the IV conditions.

Univariate statistics for each of the DVs, for each IV condition.

Output:

Cell Means and Standard Deviations				
Variable ..		Mean	Std. Dev.	N
INTERERR	errors during intermittent noise condition			
For entire sample		20.000	3.162	6
CONSTERR	errors during constant noise condition			
For entire sample		28.167	3.488	6
INTERSPD	speed during intermittent noise condition			
For entire sample		114.500	10.986	6
CONSTSPD	speed during constant noise condition			
For entire sample		112.000	16.310	6

EFFECT .. NOISE
 Multivariate Tests of Significance (S = 1, M = 0, N = 1)

Test Name	Value	Approx. F	Hypoth. DF	Error DF	Sig. of F
Pillais	.89246	16.59763	2.00	4.00	.012
Hotellings	8.29881	16.59763	2.00	4.00	.012
Wilks	.10754	16.59763	2.00	4.00	.012
Roys	.89246				

Multivariate significance test. SPSS provides four multivariate tests (three with an approximate F-value and associated p-value). With two IV conditions and equal sample sizes, these F-approximations will be equal. Usually they produce equivalent decisions about whether to reject or retain H0: . Wilks is probably the most commonly reported multivariate summary statistic.

Based on these results we would reject the multivariate H0: and conclude that there is a multivariate mean difference involving these DVs, between these two IV conditions.

SPSS Code:

```
manova intererr consterr
  / wsfactors noise(2)
  / print signif(avonly).
```

Having found a multivariate effect, we then want to examine each of the DVs for a mean difference between the IV conditions. The first analysis examines the error DV for an effect; the second examines the speed DV.

Output:

Tests involving 'NOISE' Within-Subject Effect.

AVERAGED Tests of Significance for MEAS.1 using UNIQUE sums of squares

Source of Variation	SS	DF	MS	F	Sig of F
WITHIN CELLS	34.42	5	6.88		
NOISE	200.08	1	200.08	29.07	.003

We would conclude that there is a mean difference between the mean error rates during intermittent and constant noise.

Reporting the Results

Task performance under the different noise conditions is summarized in Table 1. There was a multivariate difference between performance in the intermittent noise and constant noise conditions (Wilks = .108, F(2,4) = 16.69, p = .012). As hypothesized, there were fewer errors committed during the intermittent noise condition than during the constant noise condition (F(1,5) = 29.07, Mse = 6.88, p = .003). However, contrary to the research hypothesis, there was no difference between the mean task completion speeds for the two noise conditions (F(1,5) = .68, Mse = 27.55, p = .447).

SPSS Code:

```
manova interspd constspd
  / wsfactors noise(2)
  / print signif(avonly).
```

Output:

Tests involving 'NOISE' Within-Subject Effect.

AVERAGED Tests of Significance for MEAS.1 using UNIQUE sums of squares

Source of Variation	SS	DF	MS	F	Sig of F
WITHIN CELLS	137.75	5	27.55		
NOISE	18.75	1	18.75	.68	.447

Table 1

Mean (stdev) performance scores under the noise conditions.

Performance Measure	Noise Condition	
	Intermittent Noise	Constant Noise
Number of Errors	20.00 (3.16)	28.17 (3.49)
Speed of Completion	114.50 (10.99)	112.00 (16.31)

We would conclude that there is no mean difference between the speed with which that task was completed during intermittent and constant noise conditions.